

الفصل الرابع
صيغ المصفوفات ARRAYS FORMULAS

4. صيغ المصفوفات

المصفوفات من أهم المزايا التي يقدمها الاكسيل فبواسطتها يمكن انشاء صيغ معقدة يصعب انشاؤها من خلال المعادلات العادية. والمصفوفة هي عبارة عن مجموعة من القيم التي يتم التعامل معها بشكل افرادي أو إجمالي وقد تكون المصفوفة وحيدة البعد أي أن القيم مدخلة في صف أو عمود أو ثنائية الأبعاد حيث تكون المصفوفة مدخلة في مجموعة من الصفوف والاعمدة وفي الشكل 1-4 نموذج لمصفوفة وحيدة البعد ومصفوفة ثنائية الأبعاد

	T	S	R	Q	P	O	N	M	L
	105	78	45	23	25	45	20	12	
					12	78	89		
					48	56	14		
					96	14	125		

الشكل 1-4

و نلاحظ ان المصفوفة تكتب بشكل معين فلو نظرنا الى شريط الصيغة نجد أن قيم المصفوفة مُدخلة بداخل الأقواس المنحنية Karley brackets. وسوف نتعرف في الفقرات التالية على كيفية ادخال المصفوفات والتعامل معها.

في الصيغ العادية نحن -في الغالب- نجري العمليات على خلايا منفردة ولكن في صيغ المصفوفات فنحن نجري العمليات على نطاق من الخلايا في نفس الوقت.

صيغ المصفوفات متعددة الخلايا

يعرض الشكل 2-4 نطاق من البيانات يوضح اجمالي مبيعات عدد من المنتجات والمطلوب حساب اجمالي المبيعات لكل منتج من المنتجات. يمكن حساب القيمة في العمود D باستخدام الصيغة الموجودة في الخلية D2 ثم نسخها للأسفل على باقي الخلايا

$$=B2*C2$$

حيث سنصل في النهاية الى 6 معادلات تحسب اجمالي المبيعات لكل منتج من المنتجات

fx		=B2*C2					
G	F	E	D	C	B	A	
	المعادلة المستخدمة		الإجمالي	سعر القطعة	الوحدات المباعة	المنتج	1
	=B2*C2		\$150	\$50	3	AR-998	2
	=B3*C3		\$1,000	\$100	10	BZ-011	3
	=B4*C4		\$100	\$20	5	MR-919	4
	=B5*C5		\$90	\$10	9	TR-811	5
	=B6*C6		\$180	\$60	3	TS-333	6
	=B7*C7		\$200	\$200	1	ZL-001	7
							8
							9
						اجمالي المبيعات لكل الأصناف	10
							11
						اجمالي المبيعات لكل الأصناف (صيف)	12
							13

الشكل 2-4

وهناك طريقة بديلة يتم فيها استخدام صيغة وحيدة (صيغة مصفوفة) لحساب جميع القيم في النطاق D2:D7. تشغل هذه الصيغة الوحيدة 6 خلايا و ترجع صفيفا من 6 قيم.

ولإنشاء صيغة صفيف وحيدة لتنفيذ عملية الحساب نتبع الخطوات التالية:

1. حدد النطاق الذي سيحتوي على النتائج في هذا المثال D2:D7.
2. اكتب الصيغة التالية:

$$=B2:B7*C2:C7$$

3. عند ادخال صيغة عادية فإننا نضغط على مفتاح الادخال Enter ولكن لإدخال صيغة مصفوفة نضغط CTRL+SHIFT+ENTER.

يقوم الاكسيل بإدراج الصيغة في جميع الخلايا المحددة. وإذا تفحصت الصيغة في شريط الصيغة فإنك ستجد ما يلي:

$$\{=B2:B7*C2:C7\}$$

الشكل 4-3

	G	F	E	D	C	B	A	
1		المعادلة المستخدمة		الإجمالي	سعر القطعة	الوحدات المباعة	المنتج	1
2		{=B2:B7*C2:C7}		\$150	\$50	3	AR-998	2
3		{=B2:B7*C2:C7}		\$1,000	\$100	10	BZ-011	3
4		{=B2:B7*C2:C7}		\$100	\$20	5	MR-919	4
5		{=B2:B7*C2:C7}		\$90	\$10	9	TR-811	5
6		{=B2:B7*C2:C7}		\$180	\$60	3	TS-333	6
7		{=B2:B7*C2:C7}		\$200	\$200	1	ZL-001	7
8								8
9								9
10							اجمالي المبيعات لكل الأصناف	10
11								11
12							اجمالي المبيعات لكل الأصناف (صفيف)	12

يقوم الاكسيل بإدخال الاقواس المنحنية للدلالة على ان هذه صيغة صفيف.

تقوم هذه الصيغة بتنفيذ عمليات الحساب و تعيد مصفوفة مكونة من 6 عناصر. تعمل الصيغة عمليا مع اثنتين من المصفوفات الأخرى مخزنين في نطاقين مختلفين حيث يحتوي النطاق B2:B7 قيم الصفيف الأول و النطاق C2:C7 عناصر الصفيف الثاني.

وبما أن اظهر أكثر من قيمة واحدة في نفس الخلية غير ممكن فإنك بحاجة الى 6 خلايا لإظهار الناتج. وهذا يفسر تحديد 6 خلايا قبل ادخال صيغة المصفوفة.

مزايا استخدام صيغة مصفوفة واحدة بدلا من الصيغ المفردة :

1. طريق جيدة لضمان ان جميع الصيغ في النطاق متماثلة.
2. هذا الامر يقلل من احتمال الكتابة فوق الصيغة بشكل غير مقصود. اذا انه لا يمكن تغيير خلية واحدة في صيغة الصفيف متعددة الخلايا.
3. يمنع المبتدئين من العبث أو التلاعب بالصيغ.

صيغ المصفوفات وحيدة الخلية

لنأخذ المثال التالي كمثال على صيغ المصفوفات وحيدة الخلية (استخدمنا نفس بيانات المثال السابق)

{=SUM (B2:B7*C2:C7)}

يمكن ادخال هذه الصيغة في أي خلية، تذكر أن تضغط Ctrl+Shift+Enter وألا تكتب الأقواس المنحنية.

تعيد صيغة الصفيف السابقة مجموع المبيعات لكافة المنتجات، يجب أن تتأكد أن هذه الصيغة لا تعتمد على البيانات الموجودة في العمود D. تعمل هذه الصيغة مع صفيفين من الخلايا الصفيف الأول في النطاق B2:B7 والثاني C2:C7. تقوم هذه الصيغة بضرب القيم المتوافقة في الصفيفين وانشاء صفيف جديد موجود في الذاكرة فقط. تقوم الدالة SUM بعد ذلك بالتعامل مع هذا الصفيف و ترجع مجموع كافة عناصره.

المصفوفات الثابتة

في المثال السابق استخدمنا مصفوفات مخزنة في نطاقات، ما نريد أن نوضحه هنا هو أنه ليس من الضروري تخزين المصفوفة في نطاق من الخلايا. فيمكن انشاء وتخزين المصفوفات الثابتة في الذاكرة.

يمكن انشاء المصفوفة الثابتة بسرد عناصرها ووضعها بين قوسين منحنين {}, فيما يلي مثال عن مصفوفة ثابتة:

{10,0,5,25}

فيما يلي مثال عن صيغة تستخدم اثنين من المصفوفات الثابتة:

=SUM({1,2,3,4}*{5,6,7,8})

تقوم هذه الصيغة بإنشاء مصفوفة جديدة يتم تخزينها في ذاكرة الحاسوب RAM تتكون هذه المصفوفة من حاصل ضرب القيم المتقابلة في المصفوفتين وتكون قيمها كالتالي:

{5,12,21,32}

ثم تستخدم الدالة SUM هذه المصفوفة كوسيط وتعيد القيمة 70.

ملاحظة:

عند كتابة المصفوفة بشكل مباشر (كما في الصيغة السابقة) فإن عليك كتابة القوسين المنحنيين {} حول عناصر المصفوفة. ولكن عند ادخال صيغة مصفوفة يجب عليك ألا تكتب هاته الأقواس وإنما تضغط **Ctrl+Shift+Enter**

أبعاد المصفوفات

ذكرنا فيما سبق أن المصفوفة قد تكون وحيدة البعد أو ثنائية الأبعاد. بالنسبة للمصفوفة وحيدة البعد، يمكن أن يتم تخزينها أفقياً أو عمودياً.

المصفوفة وحيدة البعد الأفقية

يتم الفصل بين عناصر المصفوفة وحيدة البعد الأفقية باستخدام الفاصلة (,) وفيما يلي مثال عن مصفوفة وحيدة البعد أفقية:

{40,5,45,12}

ويتطلب إظهار هذه المصفوفة تحديد نطاق مكون من أربع خلايا متجاورة في صف واحد ومن ثم كتابة الصيغة التالية:

={40,5,45,12}

ثم الضغط على Ctrl+Shift+Enter.

المصفوفة وحيدة البعد الرأسية

يتم الفصل بين عناصر المصفوفة وحيدة البعد الرأسية باستخدام الفاصلة المنقوطة (;) وفيما يلي مثال عن المصفوفة الرأسية وحيدة البعد:

{12;45;78;50}

ويتطلب اظهار هذه المصفوفة تحديد نطاق مكون من أربع خلايا متجاورة في عمود واحد ومن ثم كتابة الصيغة التالية:

={12;45;78;50}

ثم الضغط على Ctrl+Shift+Enter.

المصفوفات ثنائية البعد

في المصفوفات ثنائية البعد يتم الفصائل بين العنصر الأفقية باستخدام الفاصلة (,) وبين العنصر الرأسية باستخدام الفاصلة المنقوطة (:). يعرض المثال التالي مصفوفة مكونة من 3 صفوف و 4 أعمدة:

{1,2,3;4,5,6;7,8,9;10,11,12}

ويتطلب إظهار هذه المصفوفة تحديد نطاق مكون من ثلاث صفوف وأربع أعمدة ثم ادخال الصيغة التالية:

={1,2,3;4,5,6;7,8,9;10,11,12}

ثم نضغط على المفاتيح Ctrl+Shift+Enter.

يعرض الشكل 4-4 هذه المصفوفة في النطاق D3:F9

	D	E	F	G
1				
2				
3	1	2	3	
4	4	5	6	
5	7	8	9	
6	10	11	12	

الشكل 4-4

إنشاء مصفوفة من القيم في نطاق معين

في الشكل 4-5 تقوم صيغة المصفوفة التالية والمدخلة في النطاق K10:M12 بإنشاء مصفوفة انطلاقاً من نطاق بيانات موجود بداخل ورقة العمل.

{=K5:M7}

الشكل 4-5

	K	L	M
4			
5	12	75	89
6	45	102	45
7	78	23	11
8			
9			
10	12	75	89
11	45	102	45
12	78	23	11
13			

إن المصفوفة في النطاق K10:M12 مرتبطة بالخلايا في النطاق K5:M7 ويؤدي تغيير أي خلية في النطاق K5:M7 إلى تغيير الخلية الموافقة في النطاق K10:M12 .

إنشاء مصفوفة ثابتة من القيم في نطاق محدد

في المثال السابق قامت الصيغة في الخلايا K10:M12 بإنشاء ارتباط إلى الخلايا في النطاق K5:M7 ولقطع هذا الارتباط وإنشاء مصفوفة ثابتة عناصرها مكونة من القيم في النطاق K5:M7 نحدد خلايا النطاق K10:M12 ، نضع مؤشر الماوس في شريط الصيغة لتحرير الصيغة الموجودة ثم نضغط على زر F9 لتحويل الخلايا إلى قيم ثابتة، ثم نضغط Ctrl+Shift+Enter لإعادة ادخال المصفوفة التي أصبحت تستخدم قيماً ثابتة.

أمثلة على استخدام المصفوفات وحيدة الخلية

حساب عدد الأحرف في نطاق معين

في الشكل 4-6 لدينا نطاق من الخلايا التي تحتوي على ادخالات نصية فاذا أردنا معرفة العدد الكلي للأحرف في ذلك النطاق فان الطريقة التقليدية هي انشاء الصيغة التالية في الخلية B1 ونسخها للأسفل

=LEN(A1)

ثم استخدام دالة SUM لحساب مجموعة القيم الي تعيدها الصيغ المرحلية السابقة.

وبالإمكان الحصول على نفس النتيجة دون الحاجة لأي صيغ مرحلية من خلال معادلة المصفوفة التالية:

{=SUM(LEN(A1:A14))}

تستخدم هذه الصيغة دالة LEN لإنشاء مصفوفة جديدة يتم تخزينها في الذاكرة فقط ، قيم هذه المصفوفة هي عبارة عن عدد الأحرف الموجودة في كل خلية من خلايا النطاق. وفي هذا المثال فإن المصفوفة الجديدة سوف تكون كالتالي:

{2;3;2;5;4;4;5;4;5;5;4;5;5;5}

ومن ثم سوف تصبح صيغة المصفوفة كالتالي:

{=SUM({5;5;5;4;5;5;4;5;4;4;5;2;3;2})}

B	A
	سيارة 1
	طيارة 2
	سفينة 3
	عربة 4
	شاحنة 5
	دولاب 6
	محور 7
	عبارة 8
	مركب 9
	زورق 10
	دراجة 11
	بر 12
	بحر 13
	جو 14
	15
58	إجمالي الأحرف 16
	17

الشكل 4-6

جمع القيم الثلاث الصغرى (أو الكبرى) في نطاق

تعيد الصيغة التالية مجموع القيم الثلاث الصغرى في النطاق A1:A10 والمسمى بالبيانات:

{=SUM(SMALL(البيانات){1,2,3}))}

تستخدم هذه الصيغة صفيفاً ثابتاً كوسيط ثانٍ للدالة SMALL. يتم تقييم الدالة SMALL ثلاث مرات. وفي كل مرة يتم استخدام أحد عناصر الصفيغ الثابت كوسيط ثانٍ للدالة. في المرة الأولى تكون قيمة الوسيط 1 وترجع الدالة SMALL القيمة -5. وفي المرة الثانية تكون قيمة الوسيط الثاني هي 2 وترجع الدالة القيمة 0 وفي المرة الثالثة تكون قيمة الوسيط الثاني هي 3 وترجع الدالة القيمة 2. وعلى ذلك فالمصفوفة التي يتم تمريرها إلى الدالة SUM هي:

{-5,0,2}

وتعيد الصيغة مجموع عناصر هذه المصفوفة وهو -3.

B		A	
		12	1
		-5	2
		3	3
		2	4
		0	5
		6	6
		13	7
		7	8
		4	9
		8	10
			11
			12
			13

الشكل 7-4

في هذه المعادلة؛ ناتج الدالة ISTEEXT عبارة عن مصفوفة قيمها عبارة عن TRUE أو FALSE في هذه المصفوفة سوف يتم ضرب عناصرها مع الرقم 1 فينتج مصفوفة جديدة قيمها 1 أو 0 وهذه المصفوفة سيتم جمع عناصرها للوصول للنتيجة المطلوبة.

أمثلة أخرى متقدمة على استخدام صيغ المصفوفات

هل تظهر القيمة في النطاق؟

في هذا المثال؛ المطلوب هو كتابة معادلة في الخلية D3 تحدد هل الاسم الموجود في الخلية C3 موجود ضمن نطاق البيانات المُعطى أم لا.

X ✓ fx {=IF(OR(TheName=NameList);"موجود";"غير موجود")}							
G	F	E	D	C	B	A	
			موجود	Curt		اكتب اسماً	
			Richard	Lyle	Harold	Daniel	Al
			Rick	Maggie	Ian	Dave	Allen
			Robert	Margaret	Jack	David	Andrew
			Rod	Marilyn	James	Dennis	Anthony
			Roger	Mark	Jan	Don	Arthur
			Ronald	Marvin	Jeff	Donald	Barbara
			Russ	Mary	Jeffrey	Doug	Bernard
			Sandra	Matt	Jerry	Douglas	Beth
			Scott	Mel	Jim	Ed	Bill
			Simon	Merle	Joe	Edward	Bob
			Stacy	Michael	John	Eric	Brian
			Stephen	Michelle	Joseph	Fran	Bruce
			Steven	Mike	Kathy	Frank	Cark
			Stuart	Norman	Kathy	Fred	Carl
			Susan	Patrick	Keith	Gary	Charles
			Terry	Paul	Kenneth	George	Chris
			Thomas	Peter	Kevin	Glenn	Chuck
			Timothy	Phillip	Larry	Gordon	Clark
			Vincent	Ray	Leonard	Greg	Curt

الشكل 9-4

الغاء الصيغ المرحلية

يعتبر الغاء الصيغ المرحلية في ورقة العمل من المزايا الرئيسية لاستخدام صيغ المصفوفات حيث يؤدي ذلك إلى حذف العمليات الحسابية غير الضرورية وبالتالي تصغير ورقة العمل. في الشكل 10-4 تظهر علامات الطلاب في الامتحان الأول والثاني ويحتوي العمود D على الصيغ التي تحسب التغيرات بين علامات الامتحانيين وتحتوي الخلية D17 الصيغة التالية التي تحسب متوسط التغير لكافة الطلاب:

=AVERAGE(D2:D15)

وباستخدام صيغ المصفوفات يمكن التخلص من الصيغ في العمود D. تقوم صيغة المصفوفة التالية بحساب متوسط التغير دون الحاجة للصيغ في العمود D:

{=AVERAGE(C2:C15-B2:B15)}

حيث أن هذه الدالة تستخدم صيغتين يمثلان النطاقين B2:B15 و C2:C15 وتقوم بإنشاء مصفوفة جديدة تكون عناصرها عبارة عن الفروقات بين عناصر المصفوفتين السابقتين، ويتم تخزين هذه المصفوفة في ذاكرة الحاسوب ثم تستخدم الدالة AVERAGE هذه المصفوفة الجديدة كوسيط وتعيد الناتج.

D	C	B	A
الفرق	الامتحان الثاني	الامتحان الأول	الطالب
11		67	56
15		74	59
(6)		92	98
1		79	78
19		100	81
2		94	92
0		100	100
7		99	92
15		69	54
1		92	91
8		88	80
23		68	45
21		92	71
(11)		83	94
7.57	متوسط التغير (معادلة عادية):		
7.57	متوسط التغير (صيف):		

الشكل 10-4

إيجاد مجموع مبيعات شهر معين

في هذا المثال لدينا نطاق من البيانات يوضح المبيعات التي تمت في أيام محددة والمطلوب هو إيجاد مجموع المبيعات لشهر محدد. الفكرة ستكون كالتالي: سوف ننشئ مصفوفة في الذاكرة تكون عناصرها عبارة عن المبيعات الموافقة للأيام الموجودة ضمن نطاق الشهر المطلوب (فعلى سبيل المثال؛ لو كان الشهر المطلوب هو يناير فعناصر هذه المصفوفة سوف تكون 1500،680،369) ومن ثم سوف نجمع عناصر هذه المجموعة للحصول على مجموع المبيعات للشهر المطلوب. فالمعادلة سوف تكون كالتالي:

$$\{=SUM(IF(MONTH(B4:B13)=H4;C4:C13;0))\}$$

حيث أن الشهر المطلوب إيجاد مجموع مبيعاته تم إدخاله في الخلية H4. وهنا نحن استخدمنا دالة MONTH للحصول على الشهر الموافق للتاريخ المعطى (لاحظ أن معامل الدالة MONTH عبارة عن نطاق من البيانات وليس خلية واحدة وبالتالي ناتج تنفيذ هذه الدالة هو عبارة عن مصفوفة من القيم) ومن ثم استخدمنا دالة IF لأرجاع قيمة مبيعات اليوم الذي يقع ضمن الشهر المحدد ووضعها كعنصر ضمن مصفوفة جديدة (في الذاكرة) حيث أن الدالة IF تقارن عناصر المصفوفة التي تم الحصول عليها من خلال الدالة MONTH مع محتوى الخلية H4 وعند وجود تطابق فإنها ترجع العنصر الموافق في نطاق المبيعات C4:C13. وفي النهاية يتم جمع عناصر هذه المصفوفة الأخيرة من خلال دالة SUM.

الشكل 11-4

التاريخ	المبيعات
15/01/2014	1500
20/02/2014	2400
01/01/2014	680
03/06/2014	1250
28/01/2014	369
12/11/2014	1470
06/07/2014	2356
03/02/2014	1800
06/03/2014	1700
21/03/2014	4500
	2549

حساب عدد القيم الفريدة ضمن نطاق محدد

المطلوب في هذا المثال هو حساب عدد القيم غير المتكررة في نطاق ما في نطاق البيانات الظاهر في الشكل 4-12 مجموعة من مندوبي المبيعات ومجموع المبيعات التي أتمها كل مندوب. والمطلوب هو إيجاد عدد مندوبي المبيعات. في هذه الحالة لا نستطيع استخدام دالة COUNTA للنطاق A5:A17 مثلاً لأن بعض مندوبي المبيعات تكرر أكثر من مرة (مثلاً Kamal تكرر ثلاث مرات) وبالتالي فإن استخدام الدالة COUNTA سوف ينتج عنه عدد مندوبي المبيعات المتكررين أكثر مرة وبالتالي سوف نحصل على نتيجة خاطئة.

وفكرة الحل هي كالتالي: إعطاء كل مندوب من مندوبي المبيعات وزن معين هو عبارة عن 1 تقسيم عدد مرات ظهور مندوب المبيعات في النطاق فمثلاً Kamal يظهر ثلاث مرات في النطاق فوزن Kamal لكل مرة من مرات الظهور هو $1/3 = 0.333333$ وبالتالي عند جمع الأوزان تظهر لنا النتيجة بالشكل السليم (فمثلاً عند جمع أوزان كمال سوف تكون النتيجة 1 وليس 3 وهو المطلوب). فالمعادلة المطلوبة هي كالتالي:

`{=SUM(1/COUNTIF(A5:A17;A5:A17))}`

في هذه المعادلة استخدمنا دالة COUNTIF لمعرفة عدد مرات ظهور كل مندوب من مندوبي المبيعات (دالة COUNTIF تأخذ معاملين المعامل الأول هو نطاق العد وفي حالتنا هو A5:A15 و المعامل الثاني هو المعيار الذي سيتم العد على أساسه)، ناتج تنفيذ هذه الدالة هو عبارة عن مصفوفة سوف تخزن في الذاكرة، عناصرها عبارة عن عدد مرات ظهور كل مندوب، ومن ثم سوف يتم تقسيم 1 على كل عنصر من عناصر هذه المصفوفة لإنشاء مصفوفة الأوزان التي سيتم جمع عناصرها من خلال دالة SUM للحصول على الناتج.

الشكل 4-12

fx {=SUM(1/COUNTIF(A5:A17;A5:A17))}				
D	C	B	A	
عدد مندوبي المبيعات الكلي		Total Cost	Salesperson	4
9		\$ 3,068.55	kamal	5
		\$ 681.90	mohammed	6
		\$ 559.93	kamal	7
		\$ 506.85	kamal	8
		\$ 959.88	ahmed	9
		\$10,399.35	khalid	10
		\$ 799.90	mona	11
		\$ 1,599.90	sameer	12
		\$ 2,045.70	ahmed	13
		\$ 4,432.35	hisham	14
		\$ 1,689.50	majid	15
		\$ 1,363.80	sameer	16
		\$ 1,858.45	mahmoud	17
				18

دوال المصفوفات

في الاكسيل هناك بعض الدوال الخاصة بالمصفوفات هذه الدوال تكون نتيجة تنفيذها عبارة عن مجموعة من القيم وليس قيمة محددة كباقي الدوال العادية. وفي هذا الفصل سوف نتعرف على أهم دالتين من هذه الدوال وهما دالتي FREQUENCY و TRANSPOSE.

دالة FREQUENCY

وهي تحسب عدد التكرارات في نطاق معين. وكتطبيق على استخدام دالة FREQUENCY سوف نأخذ المثال التالي:

لدينا نطاق من البيانات يوضح لنا مجموعة من الموظفين (ممثلين برقم الضمان الاجتماعي) ويقابل كل موظف الراتب الخاص به والمطلوب إيجاد أعداد الموظفين التي تقع ضمن شرائح الراتب الموضحة في النطاق D3:D11 فمثلاً في الخلية E5 يجب أن يظهر عدد الموظفين الواقعين ضمن الشريحة "أقل من 10000" بينما في الخلية E6 يجب أن يظهر عدد الموظفين الواقعين ضمن شريحة " من 10000 حتى أقل من 20000" وهكذا.

للوصول لهذه النتيجة نحدد النطاق E5:E11 ثم نكتب المعادلة التالية:

`{=FREQUENCY(B:B;D3:D11)}`

ومن ثم نضغط CTRL+SHIFT+ENTER

E3		fx		{=FREQUENCY(B:B;D3:D11)}	
	A	B	C	D	E
1	SS #	Salary			
2	201-00-1749	37,840		Range	Frequency
3	201-74-0059	10,000		10,000	5
4	209-14-0567	15,744		20,000	31
5	215-88-1661	50,840		30,000	77
6	219-36-3581	71,670		40,000	92
7	220-45-2644	31,970		50,000	107
8	221-28-2117	31,910		60,000	55
9	224-12-6997	46,220		70,000	101
10	224-76-5527	54,230		80,000	91
11	224-92-9080	72,700		90,000	61
12	229-00-1783	32,140			
13	229-10-6118	81,930			
14	233-34-7333	78,950			
15	233-52-3221	61,890			
16	233-90-5942	25,120			
17	236-30-2115	35,680			
18	238-33-0180	47,610			
19	238-49-6767	43,580			
20	238-58-3264	78,170			
21	240-96-0244	24,840			

الشكل 13-4

دالة TRANSPOSE

دالة TRANSPOSE تعمل على تدوير الأعمدة والصفوف في نطاق محدد فهي تحول الصفوف إلى أعمدة والعكس.

ولتدوير النطاق الظاهر في الشكل 4-15 نحدد نطاق من الخلايا الفارغة بعدد صفوف يساوي عدد أعمدة النطاق الأصلي (في حالتنا النطاق الأصلي هو A3:I6) وعدد أعمدة يساوي عدد صفوف النطاق الأصلي (في هذا المثال المطلوب تحديد نطاق مكون من أربع أعمدة و تسع صفوف) ومن ثم نكتب المعادلة التالية:

{=TRANSPOSE(A3:I6)}

ثم نضغط CTRL+SHIFT+ENTER فتظهر لنا النتيجة كما في الشكل 4-16

	Jan	Feb	Mar	Apr	May	Jun	Total	Average
Sales	\$ 155.00	\$ 180.00	\$ 250.00	\$ 240.00	\$ 300.00	\$ 450.00	\$ 1,575.00	\$ 262.50
Expenses	200.00	130.00	120.00	220.00	260.00	350.00	1,280.00	213.33
Profits	(45.00)	50.00	130.00	20.00	40.00	100.00	295.00	49.17


الشكل 4-15

الشكل 4-16

	Jan	Feb	Mar	Apr	May	Jun	Total	Average
Sales	\$ 155.00	\$ 180.00	\$ 250.00	\$ 240.00	\$ 300.00	\$ 450.00	\$ 1,575.00	\$ 262.50
Expenses	200.00	130.00	120.00	220.00	260.00	350.00	1,280.00	213.33
Profits	(45.00)	50.00	130.00	20.00	40.00	100.00	295.00	49.17
Average	262.50	213.3333	49.16667					

ولاحظ أن القيم الموجودة في نطاق الناتج مرتبطة بالقيم الموجودة في النطاق الأصلي فلا يمكن إجراء أي تعديل على خلايا نطاق الناتج، بينما إذا تم أي تعديل على النطاق الأصلي ينعكس هذا التعديل على النطاق الناتج.

من الجدير بالذكر أنه يمكن إجراء عملية التدوير بدون معادلات من خلال ميزة اللصق الخاص حيث أنه بالإمكان تحديد النطاق المراد تدويره ثم اختيار نسخ Copy ومن ثم في المكان المطلوب إظهار النتيجة به نضغط بزر الماوس الأيمن ونختار Paste Special لصق خاص ثم نحدد الخيار Transpose تدوير. وفي هذه الحالة لا تكون خلايا الناتج مرتبطة بخلايا الأصل.



الشكل 17-4